NAG Fortran Library Routine Document

M01DKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

M01DKF ranks the columns of a matrix of integer numbers in ascending or descending order.

2 Specification

```
SUBROUTINE M01DKF(IM, LDM, M1, M2, N1, N2, ORDER, IRANK, IFAIL)
INTEGER           IM(LDM,N2), LDM, M1, M2, N1, N2, IRANK(N2), IFAIL
CHARACTER*1       ORDER
```

3 Description

M01DKF ranks columns N1 to N2 of a matrix, using the data in rows M1 to M2 of those columns. The ordering is determined by first ranking the data in row M1, then ranking any tied columns according to the data in row M1 + 1, and so on up to row M2.

M01DKF uses a variant of list-merging, as described by Knuth (1973), pp 165-166. The routine takes advantage of natural ordering in the data, and uses a simple list insertion in a preparatory pass to generate ordered lists of length at least 10. The ranking is stable: equal columns preserve their ordering in the input data.

4 References

Knuth D E (1973) *The Art of Computer Programming (Volume 3)* (2nd Edition) Addison-Wesley

5 Parameters

1: IM(LDM,N2) – INTEGER array *Input*

On entry: rows M1 to M2 of columns N1 to N2 of IM must contain integer data to be ranked.

2: LDM – INTEGER *Input*

On entry: the first dimension of the array IM as declared in the (sub)program from which M01DKF is called.

Constraint: LDM $\geq$ M2.

3: M1 – INTEGER *Input*

On entry: the index of the first row of IM to be used.

Constraint: M1 $>$ 0.

4: M2 – INTEGER *Input*

On entry: the index of the last row of IM to be used.

Constraint: M2 $\geq$ M1.

5: N1 – INTEGER *Input*

On entry: the index of the first column of IM to be ranked.

Constraint: N1 > 0.

6: N2 – INTEGER *Input*

On entry: the index of the last column of IM to be ranked.

Constraint: N2 $\geq$ N1.

7: ORDER – CHARACTER*1 *Input*

On entry: if ORDER is 'A', the columns will be ranked in ascending (i.e., non-decreasing) order; if ORDER is 'D', into descending order.

Constraint: ORDER $=$ 'A' or 'D'.

8: IRANK(N2) – INTEGER array *Output*

On exit: elements N1 to N2 of IRANK contain the ranks of the corresponding columns of IM. Note that the ranks are in the range N1 to N2: thus, if the ith column of IM is the first in the rank order, IRANK(i) is set to N1.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, M2 < 1,
or N2 < 1,
or M1 < 1,
or M1 $>$ M2,
or N1 < 1,
or N1 $>$ N2,
or LDM $<$ M2.

IFAIL $= 2$

On entry, ORDER is not 'A' or 'D'.

7 Accuracy

Not applicable.

8 Further Comments

The average time taken by the routine is approximately proportional to $n \times \log n$, where $n = \mathrm{N2} - \mathrm{N1} + 1$.

9 Example

The example program reads a matrix of integers and ranks the columns in descending order.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     M01DKF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          MMAX, NMAX
      PARAMETER        (MMAX=20,NMAX=20)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N
*     .. Local Arrays ..
      INTEGER          IM(MMAX,NMAX), IRANK(NMAX)
*     .. External Subroutines ..
      EXTERNAL         M01DKF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'M01DKF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.GE.1 .AND. M.LE.MMAX .AND. N.GE.1 .AND. N.LE.NMAX) THEN
         DO 20 I = 1, M
            READ (NIN,*) (IM(I,J),J=1,N)
   20    CONTINUE
         IFAIL = 0
*
         CALL M01DKF(IM,MMAX,1,M,1,N,'Descending',IRANK,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Data'
         WRITE (NOUT,*)
         DO 40 I = 1, M
            WRITE (NOUT,99999) (IM(I,J),J=1,N)
   40    CONTINUE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Ranks'
         WRITE (NOUT,*)
         WRITE (NOUT,99999) (IRANK(I),I=1,N)
      END IF
      STOP
*
99999 FORMAT (1X,12I6)
      END
```

9.2 Program Data

```
M01DKF Example Program Data
3 12
5 4 3 2 2 1 9 4 4 2 2 1
3 8 2 5 5 6 9 8 9 5 4 1
9 1 6 1 2 4 8 1 2 2 6 2
```

9.3 Program Results

```
M01DKF Example Program Results

Data

     5     4     3     2     2     1     9     4     4     2     2     1
     3     8     2     5     5     6     9     8     9     5     4     1
     9     1     6     1     2     4     8     1     2     2     6     2

Ranks

     2     4     6     9     7    11     1     5     3     8    10    12
```